


Complex Custom Clinical Graphs Step by Step with SAS® ODS Statistical Graphics

Richann Watson is an independent statistical programmer and CDISC consultant based in Ohio who loves to code and is very active in the SAS User Group community. When Richann is not busy coding or volunteering in the SAS User Group community, she is spending time with her family and cute but psycho puppy, Loki, or doing some of her favorite crafts such as crocheting or sewing.

Josh Horstman is an independent statistical programming consultant and trainer based in Indianapolis with 25 years of experience using SAS, primarily in the life sciences industry. Josh is a SAS Certified Advanced Programmer who loves coding and presenting at PharmaSUG and other industry conferences. Josh also enjoys travelling and hiking with his family and has been to 47 states and 28 national parks.

Agenda

- **Overview of ODS Statistical Graphics**
- **Create a Basic Bar Chart**
 - SGPLOT
 - GTL
- **Add Titles and Footnotes to Graph Area**
 - SGPLOT
 - GTL
- **Embedding a Table Within a Graph**
 - SGPLOT
 - GTL
- **Adding a Table Outside the Graph**
 - SGPLOT
 - GTL



Quick Overview of ODS Statistical Graphics

Overview: The Output Delivery System (ODS)

- Prior to ODS, SAS limited to text-based “SAS listing” output
- ODS output makes use of colors, fonts, graphics, and more!
- ODS provides ability to produce output in various formats:



... and more!

- ODS is part of the Base SAS product since version 7 (No separate license required)

Overview: ODS Statistical Graphics

- An extension to ODS used to create analytical graphs
- Introduced in SAS 9.2 as part of SAS/GRAPH (experimental in v9.1)
- Moved into the Base SAS product in version 9.3
- Based on the Graph Template Language (GTL)

ODS Statistical Graphics – Components

- Graph Template Language (GTL)
 - comprehensive language for creating statistical graphics
- ODS Graphics procedures
 - provide a procedural interface to most common features of GTL
- ODS GRAPHICS statement
 - controls various graphic-related settings and options
- ODS Graphics Editor
 - interactive tool for modifying graphs
- ODS Graphics Designer
 - graphical interface for designing graphs

ODS Statistical Graphics – Procedures

- SGPLOT – single-cell plots
- SG PANEL – multiple-panel plots
- SGSCATTER – advanced scatter plots
- SGRENDER – render graphs written in GTL
- SGDESIGN – used with ODS Graphics Designer

Statistical Graphics vs. Legacy SAS/GRAPH

SG Procedures

- SGPLOT, SGPANEL, SGSCATTER, etc.
- Based on templates
- Creates image files
- Use ODS GRAPHICS statement to control environment
- Visual properties are set within the procedure

SAS/GRAPH


- GPLOT, GCHART, GSLIDE, GBARLINE, GCONTOUR, etc.
- Based on device drivers
- Creates catalog entries
- Use GOPTIONS statement to control environment
- Many properties set with global statements such as AXIS, LEGEND, SYMBOL, etc.

About ODS Destinations

- To create ODS graphs, a valid ODS destination must be open.
- Build an “ODS sandwich” around your graph code.
- For example, to output a graph to the PDF destination:





```
ods pdf file="c:\example.pdf";  
    <SG procedure code goes here...>;  
ods pdf close;
```

- Similar syntax for ODS HTML, ODS RTF, etc.




Creating a Simple Bar Chart

Understanding Drawspace



Bar Chart by Treatment for Percent of Patients with Dermatologic Event - 1a

Header



Footer

Bar Chart Component Only

Step 1: Creating a Simple Bar Chart

Basic SGPLOT Syntax

```
proc sgplot data=<input-data-set> <options>;  
    <one or more plot requests>  
    <other optional statements>  
run;
```

Other optional statements control specific graph features – XAXIS, YAXIS, REFLINE, INSET, KEYLEGEND, etc.

There are dozens of plot request statements available – SCATTER, SERIES, VBOX, VBAR, HIGHLOW, BUBBLE, etc.

The VBAR Statement

- Creates a vertical bar chart.

```
proc sgplot data=<input-data-set> <options>;  
    vbar categorical-variable < / options>;  
run;
```

- RESPONSE= option specifies response variable to control length of bars. (Otherwise, bars represent frequency counts.)
- STAT= option specifies statistic for length of bars (Default is SUM when RESPONSE variable is included, FREQ otherwise.)
- HBAR statement is analogous for horizontal bar charts.

Exercise 1a - Using SGPLOT

```
title "Bar Chart by Treatment for Percent of Patients with Dermatologic  
Event - &pgmname";  
footnote "SGPLOT - Bar Chart Component Only";  
  
proc sgplot data = OUTD.TRTPCT;  
  format TRTAN trt. PCT_ROW pctfmt.;  
  xaxis type = discrete label = " ";  
  yaxis type = linear  
    label = "Percentage of Patients with Dermatologic Event (%)"  
    values = (0 to 100 by 25);  
  vbar TRTAN / response = PCT ROW datalabel = PCT ROW;  
run;
```

Understanding GTL

```
proc template;  
  define statgraph recrgrphb;  
    begingraph / border = false;  
      layout overlay / xaxisopts = (label = " " type = discrete)  
        yaxisopts = (label = "Percentage of Patients with  
Dermatologic Event (%)")  
          linearopts = (tickvaluesequence = (start = 0 end = 100  
                                           increment = 25)  
                        viewmax = 100));  
      barchart x = TRTAN y = PCT_ROW / orient = vertical  
        barlabel = true;  
    endlayout;  
  endgraph;  
  end;  
run;
```

Define the structure of the graph.
The template name is used when rendering the graph.

Define statement has a corresponding END.

Understanding GTL

```
proc template;  
  define statgraph recrgrphb;  
    begingraph / border = false;  
    layout overlay / xaxisopts = (label = " " type = discrete)  
      yaxisopts = (label = "Percentage of Patients with  
Dermatologic Event (%)")  
        linearopts = (tickvaluesequence = (start = 0 end = 100  
                                          increment = 25)  
                      viewmax = 100));  
    barchart x = TRTAN y = PCT_ROW / orient = vertical  
      barlabel = true;  
    endlayout;  
    endgraph;  
  end;  
run;
```

Each STATGRAPH has at most one **BEGINGRAPH**, which signals the start of the graph template definition.

BEGINGRAPH has a corresponding **ENDGRAPH**, which signals the end of the graph template definition.

Understanding GTL

```
proc template;  
  define statgraph recrgphb;  
    begingraph / border = false;  
      layout overlay / xaxisopts = (label = " " type = discrete)  
        yaxisopts = (label = "Percentage of Patients with  
Dermatologic Event (%)")  
          linearopts = (tickvaluesequence = (start = 0 end = 100  
            increment = 25)  
            viewmax = 100));  
      barchart x = TRTAN y = PCT_ROW / orient = vertical  
        barlabel = true;  
    endlayout;  
  endgraph;  
end;  
run;
```

LAYOUT allows you to specify the type of layout. Some layouts can be nested.

For each LAYOUT, you need to signal the end of the layout with ENDLAYOUT.

Exercise 1b - Using GTL

```
proc template;
  define statgraph recrgrphb;
    begingraph / border = false;
      layout overlay / xaxisopts = (label = " " type = discrete)
        yaxisopts = (label = "Percentage of Patients with
Dermatologic Event (%)")
          linearopts = (tickvaluesequence = (start = 0 end = 100
increment = 25)
viewmax = 100));
      barchart x = TRTAN y = PCT_ROW / orient = vertical
        barlabel = true;
    endlayout;
  endgraph;
end;
run;
```

Specify the plot statement within the OVERLAY layout. Each plot statement will have their own set of options.

Exercise 1b - Using GTL

```
title "Bar Chart by Treatment for Percent of Patients with Dermatologic  
Event - &pgmname";  
footnote "GTL - Bar Chart Component Only";
```

```
proc template;
```

```
...
```

```
run;
```

```
proc sgrender data = _____ template = _____ ;
```


```
format TRTAN trt. PCT_ROW pctfmt.;
```

```
run;
```

Using SGENDER you can specify the data that will be used with the template that was defined.




Titles and Footnotes in Graph Area



Bar Chart by Treatment for Percent of Patients with Dermatologic Event - 1b

Header

Patients with Dermatologic Events



Subjects only counted once in each treatment group

Footer

Titles/Footnotes Inside and Outside Graph Area

Step 2: Titles and Footnotes in Graph Area

SG Annotations

- Allow you to add text, shapes, images, and other annotations to graph output.
- Two-step process:
 - Create an annotation dataset
 - Use the annotation dataset in the SGPLOT, SGPANEL, or SGSCATTER procedures.
- Each row in the annotation dataset represents a function.
- As of SAS 9.4, there are 10 functions: ARROW, IMAGE, LINE, OVAL, POLYCONT, POLYGON, POLYLINE, RECTANGLE, TEXT, and TEXTCONT

Using SGPLOT

Create Annotate Data Set

```
data insidetf;
  retain function 'text' drawspace 'graphpercent' width 100;
  length anchor $6 textstyleelement $17 label $51;
  input x1 y1 anchor $ textstyleelement $ label $ 32 - 82;
cards;
50 99 top      GraphTitleText      Patients with Dermatologic Events
20  1  bottom GraphFootnoteText Subjects only counted once in each treatment
group.
;
run;
```


Exercise 2a - Using SGPLOT SGANNO

```
title "Bar Chart by Treatment for Percent of Patients with Dermatologic  
Event - &pgmname";  
footnote "SGPLOT - Titles/Footnotes Inside and Outside Graph Area";  
  
proc sgplot data = OUTD.TRTPCT  
    pad = (top = 5% bottom = 5%) sganno = insidetf;  
    format TRTAN trt. PCT_ROW pctfmt.;  
    xaxis type = discrete label = " ";  
    yaxis type = linear label = "Percentage of Patients with Dermatologic  
Event (%)"  
        values = (0 to 100 by 25);  
    vbar TRTAN / response = PCT_ROW  
        datalabel = PCT_ROW;  
run;
```

Exercise 2b - Using GTL ENTRYTITLE/ENTRYFOOTNOTE

```
proc template;  
  define statgraph recrgrphb;  
    begingraph / border = false;  
      entrytitle "Patients with Dermatologic Events";  
      entryfootnote halight = left "Subjects only counted once in each  
treatment group.";  
      layout overlay / /* axis options */;  
        barchart x = TRTAN y = PCT_POW / orient = vertical  
                barlabel = true;  
      endlayout;  
    endgraph;  
  end;  
run;
```


Must be placed directly inside BEGINGRAPH block.
Displays a title above the wall area within the graph area

Embedding a Table



Bar Chart by Treatment for Percent of Patients with Dermatologic Event - 1c

Header



Subjects only counted once in each treatment group

Footer

SGPLOT - Inset Table Component Using INSET

Step 3: Embedding a Table Using INSET

SGPLOT: The INSET Statement

Adds text within the plot.

Required argument is one or more text strings

```
inset <text string(s)> / <options>;
```

Selected options:

Inset Option	Description
POSITION=	Specifies position of legend: TOP, BOTTOM (default), LEFT, RIGHT, TOPLEFT, TOPRIGHT, BOTTOMLEFT, BOTTOMRIGHT
BORDER / NOBORDER	Specifies whether a border is drawn around the text box.
TEXTATTRS=	Specifies text attributes* of legend text

Exercise 3a - Using SGPLOT INSET

```
proc sgplot data = OUTD.TRTPCT pad = (top=5% bottom=5%) sganno = insidetf;
  /* format and axis statements */
  vbar TRTAN / response = PCT_ROW
             datalabel = PCT_ROW;
```

```
  inset "Pearson's Chi-square Test Results
        Cochran-Armitage Trend Test Results"
        " Treatment Comparison Value P-Value
        Value = &cmstat"
        " Placebo - Low Dose &valuechi054
        &pchi054 P-value = &cmpvalue"
        " Placebo - High Dose &valuechi081
        &pchi081"/
        textattrs = (size = 8pt) position = top;
```

```
run;
```


```
ods rtf close;
```

```
ods pdf close;
```

```
inset "Pearson's Chi-square Test Results
      Treatment Comparison Value P-Value Cochran-Armitage Trend Test Results"
      " Placebo - Low Dose &valuechi054 &pchi054 P-value = &cmpvalue"
      " Placebo - High Dose &valuechi081 &pchi081"/ textattrs = (size = 8pt) position = top;
```

Bar Chart by Treatment for Percent of Patients with Dermatologic Event - 1c

Header



Subjects only counted once in each treatment group.

Footer

Inset Table Component Using DRAWTEXT

Step 3: Embedding a Table Using DRAWTEXT

Exercise 3b - Using GTL DRAWTEXT

```
proc template;
  define statgraph recrgrphb;
    begingraph / border = false;
      entrytitle "Patients with Dermatologic Events";
      entryfootnote halight = left "Subjects only counted once in each
treatment group.";
      layout overlay / /* axis options */;
        barchart x = TRTAN y = PCT_ROW / orient = vertical
                                barlabel = true;

        /* drawtext statements */


      endlayout;
    endgraph;
  end;
run;
```


Exercise 3b - Using GTL DRAWTEXT

```
drawtext textattrs = (size = 8pt) "Pearson's Chi-square Test Results  
Cochran-Armitage Trend Test Results"  
/ x = 20 y = 99 width = 75 widthunit = percent  
xspace = wallpercent yspace = datavalue anchor = left;  
drawtext textattrs = (size = 8pt) "Treatment Comparison  
Value P-Value Value = &cmstat"  
/ x = 20 y = 96 width = 75 widthunit = percent  
xspace = wallpercent yspace = datavalue anchor = left;  
drawtext textattrs = (size = 8pt) "Placebo - Low Dose  
&valuechi054 &pchi054 P-value =  
&cmpvalue"  
/ x = 20 y = 93 width = 75 widthunit = percent  
xspace = wallpercent yspace = datavalue anchor = left;  
drawtext textattrs = (size = 8pt) "Placebo - High Dose  
&valuechi081 &pchi081"  
/ x = 20 y = 90 width = 75 widthunit = percent  
xspace = wallpercent yspace = datavalue anchor = left;
```

Bar Chart by Treatment for Percent of Patients with Dermatologic Event - 1c

Header



Subjects only counted once in each treatment group

Footer

SGPLOT - Inset Table Component Using INSET

Step 3: Embedding a Table Using GRIDDED

Exercise 3c - Using GTL GRIDDED Layout

```
proc template;
  define statgraph recrgrphb;
    mvar valuechi054 pchi054 valuechi081 pchi081 cmstat cmpvalue;
    begingraph / border = false;
      entrytitle "Patients with Dermatologic Events";
      entryfootnote halign = left "Subjects only counted once in each
treatment group.";
      layout overlay / /* axis options */;
        barchart x = TRTAN y = PCT_ROW / orient = vertical
          barlabel = true;
        /* GRIDDED layout */
      endlayout;
    endgraph;
  end;
run;
```

Macro variables that are declared using MVAR will resolve to a string where macro variables declared with NMVAR will convert to a numeric token. With (N)MVAR the macro variable is resolved at **execution** rather than a compile time.

Using GTL GRIDDED Layout


With GRIDDED layout you specify the number of columns or rows. Default is 1 if not specified.

```
layout gridded / columns = 4 order = rowmajor autoalign = (top);
entry "Pearson's Chi-square Test Results";
entry " ";
entry " ";
entry "Cochran-Armitage Trend Test Results";
entry "    Treatment Comparison";
entry "Value";
entry "P-value";
entry "    Value = " cmstat;
entry "    Placebo - Low Dose";
entry valuechi054;
entry pchi054;
entry "P-value = " cmpvalue;
entry "    Placebo - High Dose";
entry valuechi081;
entry pchi081;
endlayout;
```

Specify the order in which the grid is filled. ROWMAJOR fills each column before moving to next row.

If a specific cell in the table should be left null, then a blank entry line should be created.

Macro variables specified with MVAR(N) are referenced. Note they are referenced without & symbol.



Adding Table Outside of Graph

Bar Chart by Treatment for Percent of Patients with Dermatologic Event - 1e

Header

Patients with Dermatologic Events

Pearson's Chi-square Test Results

Treatment Comparison

Placebo - Low Dose

Placebo - High Dose

Value

27.45

25.81

P-value


<0.001

<0.001

Cochran-Armitage Trend Test Results

Value = 5.69

P-value = <0.001



Subjects only counted once in each treatment group.

Footer

Inset Table Component Using GRIDDED

Step 4: Adding Table Outside Graph

Exercise 4 - Using GTL GRIDDED Layout with LATTICE

```
proc template;
  define statgraph recgrphb;
    mvar valuechi054 pchi054 valuechi081 pchi081 cmstat cmpvalue;
    begingraph / border = false;
      entrytitle "Patients with Dermatologic Events";
      entryfootnote halign = left "Subjects only counted once in each treatment
group.";
      layout lattice / rows = 2 rowweights = (.2 .8);

      /* GRIDDED layout */

      layout overlay / /* axis options */;
        barchart x = TRTAN y = PCT_ROW / orient = vertical barlabel = true;
      endlayout;
    endgraph;
  end;
run;
```

LATTICE allows you to split area into different sizes based on your needs. You can split into rows or columns or a combination of rows and columns.

Conclusion



What have we learned?

SGPLOT vs. GTL

SGPLOT

Advantages:

- Simple syntax
- Ideal for simple graphs
- Inset table directly into graph

Disadvantages:

- Hard to have titles and footnotes within the graph area and outside the graph area
- Can't place a table outside of the graph

GTL

Advantages:

- Embed titles and footnotes within graph area and outside the graph area
- Ideal for complex graphs
- Different techniques to inset table directly into graph

Disadvantages:

- Complex syntax

Contact Information



DATA RICH
CONSULTING

Richann Jean Watson

richann.watson@datarichconsulting.com

<http://www.datarichconsulting.com/>

<https://www.linkedin.com/in/richann-watson-31435422/>

<https://www.linkedin.com/company/datarich-consulting/>



Nested Loop
Consulting

Josh M. Horstman

josh@nestedloopconsulting.com

References and Recommended Reading

- Haworth, L. (2006). PROC TEMPLATE: The Basics. Proceedings of SUGI 31. San Francisco: SUGI. Retrieved from <https://support.sas.com/resources/papers/proceedings/proceedings/sugi31/112-31.pdf>
- Heath, D. (2018). Diving Deep into SAS® ODS Graphics Styles. Proceedings of PharmaSUG. Seattle: PharmaSUG. Retrieved from <https://www.pharmasug.org/proceedings/2018/DV/PharmaSUG-2018-DV02.pdf>
- Kriss Harris, R. W. (2020). SAS® Graphics for Clinical Trials by Example. Cary: SAS Institute Inc.
- Kuhfeld, W. (2015). Advanced ODS Graphics Examples. Cary: SAS Institute Inc. Retrieved from <https://support.sas.com/documentation/prod-p/grstat/9.4/en/PDF/odsadv.pdf>
- Kuhfeld, W. (2016). Basic ODS Graphics Examples. Cary: SAS Institute Inc. Retrieved from <http://support.sas.com/documentation/prod-p/grstat/9.4/en/PDF/odsbasicg.pdf>
- Matange, S. (2013). Getting Started with the Graph Template Language in SAS®. Cary: SAS Institute Inc.
- Matange, S. (2016). Clinical Graphs Using SAS®. Cary: SAS Institute Inc.
- Sanjay Matange, D. H. (2019). Create A Combined Graph of Tumor Data. Proceedings of SAS Global Forum. Dallas: SAS Institute Inc. Retrieved from <https://www.sas.com/content/dam/SAS/support/en/sas-global-forum-proceedings/2019/3143-2019.pdf>

References and Recommended Reading

SAS Institute Inc. (2023, Jul 19). About the Drawing Space and Drawing Units. Retrieved Dec 2023, from SAS® 9.4 and SAS® Viya® 3.5 Programming Documentation:

https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/grstatgraph/n1jn4duv8s510xn1y2nlbefm0p46.htm

SAS Institute Inc. (2023, Jul 19). DRAWTEXT Statement. Retrieved Dec 2023, from SAS® 9.4 and SAS® Viya® 3.5 Programming Documentation:

https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/grstatgraph/n0mngqrucac70n17d07ugtknu3g.htm#n1p67wdoi189fkn16c8vkq1wnqzp

SAS Institute Inc. (2023, Jul 19). ENTRYTITLE Statement. Retrieved Dec 2023, from SAS® 9.4 and SAS® Viya® 3.5 Programming Documentation:

https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/grstatgraph/n0zplmg0ard2dan17sohm8jb773n.htm

SAS Institute Inc. (2023, Jul 19). SGPLOT Procedure. Retrieved Dec 2023, from SAS® 9.4 and SAS® Viya® 3.5 Programming Documentation:

https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/grstatproc/n0yjdd910dh59zn1toodgupaj4v9.htm

Watson, R. (2019). Great Time to Learn GTL: A Step-by-Step Approach to Creating the Impossible. Proceedings of SAS Global Forum. Dallas: SAS Institute Inc. Retrieved from

<https://www.sas.com/content/dam/SAS/support/en/sas-global-forum-proceedings/2019/3170-2019.pdf>